



11/23/2019

From: *Erica Junqueira*, Graduate Student Consultant

To: *Awesome Bakery*

Re: Awesome Bakery Chocolate Chip Cookies

RESEARCH FIELD: Customer Retention

PROJECT TITLE: Chocolate Chip Cookies and Customer Satisfaction

1.0 - PROJECT DESCRIPTION

An experiment was conducted in order to measure customer satisfaction of chocolate chip cookies at Awesome Bakery, and help the bakery know how to make the cookies going forward. The experiment varied the amount of sugar and the freshness (day) of the cookies. The variables both contained 3 levels, sugar (half, regular, double) and day (1,2,3). The response variable was the taste of the cookies which was measured on a scale of 1 to 10. Each batch of cookies is the experimental unit and each cookie is the sampling unit.

There are three replications of the experiment. Nine batches of cookies were baked on the same day; three for each level of sugar. For three days, five cookies from each batch were handed out to customers, and the taste score was recorded. The experiment is a split-plot CRD design. The sugar was applied to the whole plot and each day is a split plot.

1.1 - RESEARCH QUESTIONS

Question 1: How much sugar should be in the cookies to provide high customer satisfaction?

Question 2: How long should the cookies be stored to provide high customer satisfaction?

1.2 - STATISTICAL QUESTIONS

Question 1: Are Sugar and/or Day significant variables in determining which cookie is best?

Question 2: Is there an interaction between Day and Sugar?

Question 3: How do we account for the dependent measurements of the cookies in each batch?


1.3 - VARIABLES OF INTEREST

Response variable: Taste- scale 1 to 10

Explanatory Variable- Day and Sugar

1.4 - STUDY DIAGRAM

As described above, the experiment is a split plot CRD design. There are three batches for each sugar level. The sugar is applied to the whole plot and five cookies are removed per day which represents the split plot.



2.0 - EXPLORATORY DATA ANALYSIS (EDA)

Data Cleanup

Before any statistical analysis or exploratory data analysis was conducted, data cleanup was needed. Pseudo- replication is a concern in this study because each of the cookies are not independent. Since five cookies are being tested from each batch per day (sub-plot), the scores can lead to inflated degrees of freedom. To take into account the pseudo-replication, the average was taken for the five cookies that were removed from each sub-plot. All of the EDA is conducted on the averaged data, which contains 27 observations. (See additional explanation in appendix)

The following code was used to average the data. The summarized data can be found in the appendix.

```
Proc SQL;
create table Bakery_summary as
select day, sugar, batch, avg(taste) as avg_taste
from Bakery
group by day, sugar , batch;
quit;
```

Numeric Summaries

A numeric summary table was created on the averaged data using the following code.

```
Proc Means Data=Bakery_Summary n mean std median q1 q3 min max;
class sugar day ;
var avg_taste;
Run;
```

The initial numeric summaries in table 1 show that the means for the different sugar levels on day 1 do not vary by a large amount. The regular sugar cookie performs better on day 2 than on day 1. Double sugar on day 1 and Regular sugar on day 2 have the largest mean taste score.

Table 1: Summary Table of Averaged Data

Analysis Variable : avg_taste										
sugar	day	N Obs	N	Mean	Std Dev	Median	Lower Quartile	Upper Quartile	Minimum	Maximum
double	1	3	3	6.8666667	0.5033223	6.8000000	6.4000000	7.4000000	6.4000000	7.4000000
	2	3	3	6.2666667	0.1154701	6.2000000	6.2000000	6.4000000	6.2000000	6.4000000
	3	3	3	4.8666667	0.9018500	4.8000000	4.0000000	5.8000000	4.0000000	5.8000000
half	1	3	3	6.1333333	0.9018500	6.2000000	5.2000000	7.0000000	5.2000000	7.0000000
	2	3	3	6.0666667	1.4468356	6.8000000	4.4000000	7.0000000	4.4000000	7.0000000
	3	3	3	5.0666667	0.5033223	5.0000000	4.6000000	5.6000000	4.6000000	5.6000000
regular	1	3	3	6.4000000	0.2000000	6.4000000	6.2000000	6.6000000	6.2000000	6.6000000
	2	3	3	6.8666667	0.9018500	6.8000000	6.0000000	7.8000000	6.0000000	7.8000000
	3	3	3	5.9333333	0.2309401	5.8000000	5.8000000	6.2000000	5.8000000	6.2000000

Boxplots

Boxplots were created to visualize the information that is being presented with the numeric summaries. The boxplots were created with the following SAS code.

```
Proc SGpanel data=Bakery_Summary;
  panelby Variable1 / rows=1 columns=3;
  vbox avg_taste / category= Variable2;
Run;
```

The line in middle of the boxplot represents the median and the diamond represents the mean. The averaged data was used in the creation of the boxplots. For example in figure 1, the mean and median for each days is calculated as the mean and median of the averaged scores from the 9 sub-plot of cookies per day.

Figure 1 and Figure 2 both show outliers in their boxplots. Overall there is not a large difference in the average taste score for each sugar level. Figure 2 shows a decrease in the average taste on day 3.

Figure 1: Boxplots of average taste by sugar level







Figure 2: Boxplots of average taste by day



The boxplot of average taste by batch, figure 3, reveals that there is a difference in the scores by batch. This is specifically obvious for the half sugar. Batch number 2 has a mean score that is significantly lower than the mean taste for the other 2 batches. This is an interesting observation since all other aspects of the recipe remained the same

The double sugar shows the largest decrease in the mean taste scores across the 3 days. The half sugar shows a slight decrease, but the regular sugar shows a slight increase in the mean taste score at the second day.

Figure 3: Boxplots of average taste by sugar level and batch .Figure 4: Boxplots of average taste by sugar level and day




Interaction Plots

An interaction plot was created to examine a possible interaction between sugar and day. The interaction plot was created in SAS using the following code.

```
symbol1 value=circle color=black interpol= join;
symbol2 value=square color=black interpol=join;
symbol3 value= plus color=black interpol=join;
Proc Gplot data=tasteday;
plot mn_taste*day=sugar;
Run;
```

The interaction plot reveals similar results to the box plots in figure 4. The regular sugar shows an increase in mean taste score on day 2. Double sugar shows the most significant decrease across the three days. While the behavior of the sugar levels is different for different days, it does not appear to be significant.

Figure 5: Interaction plot of average taste by sugar level and day



3.0 –STATISTICAL ANALYSIS

This chocolate chip cookie study is analyzed using analysis of variables (ANOVA). The ANOVA tests the means within each variable. If the means are significantly different, the variable will be significant at $\alpha=0.05$ significance level. As discussed earlier, pseudo-replication is a concern in this study. The ANOVA uses the summarized data containing the 27 averaged taste scores grouped by batch, sugar, and day. (See additional explanation in appendix)


Question 1: How much sugar should be in the cookies to provide high customer satisfaction?

The ANOVA was conducted using the Proc Mixed statement in SAS. The model contains Day, Batch, Sugar, and the Day*Sugar interaction. The random effect in the model is Batch nested within the different sugar levels, Batch(Sugar). This random effect serves as the denominator in the F test, used to determine the significance for Sugar.

```
Proc Mixed data= Bakery_Summarysq method=type3;
  class day batch sugar;
  model avg_taste= sugar day sugar*day /residual;
  random batch(sugar);
  store outBakery;
Run;
```

The residuals for the model were examined. The equal variance assumptions is met indicating that the model is a good fit for the data. The normality assumption has a few possible outliers, but overall the normality assumption appears to also be met. Further analysis and comparisons can be made since the assumptions are met.

Figure 6: Residual Plots



The results of the ANOVA indicate that the means for the 3 sugar levels are not significantly different as seen in the boxplots above. This concludes that the 3 different sugar levels will not have a significant influence of the taste scores of the chocolate chip cookies. It is also important to note that the Day*Sugar interaction is not significant meaning there is not a significant difference in mean sugar levels based on the different days.

Table 2: Results of ANOVA

Type 3 Tests of Fixed Effects				
Effect	Num DF	Den DF	F Value	Pr > F
sugar	2	6	1.32	0.3348
day	2	12	7.93	0.0064
day*sugar	4	12	0.95	0.4685

Significant at an $\alpha=0.05$ significance level.

Question 2: How long should the cookies be stored to provide high customer satisfaction?

Based on the results of the ANOVA in table 2, Day is significant at $\alpha=0.05$ significance level. This indicates that the means for the 3 different levels of day are not all equal which agrees with the initial findings in the EDA.

Tukey-Kramer mean comparisons were run to identify which levels of day were significantly different. The following SAS code was used to run the mean comparisons.

```
Proc Plm restore= outBakery;
    lsmeans day / adjust=tukey
    plot=meanplot cl lines;
Run;
```

The results of the mean comparisons in figure 7 show that day 1 and day 2 result in higher average taste scores than day 3. The mean taste score for day 1 and day 2 are 6.47 and 6.40 accordingly. While the mean for day 3 is 5.29. Figure 8 displays the means with the 95% confidence limits. We are 95% confident that the means are within those limits. The confidence interval for day 3 does not overlap with the confidence intervals for day 1 or day 2.

Figure 7: Tukey Mean Comparisons for Day





Figure 8: LS-Means for Day with 95% Confidence Limits.



4.0 - CONCLUSIONS

Question 1: Based on the results of the ANOVA, it can be concluded that the means for the 3 different sugar levels are not significantly different. When baking cookies for your bakery, using half, regular, or double sugar will not result in a taste score that is significantly different.

Question 2: The results of the ANOVA did conclude that the mean taste scores for the different days are significantly different. To have the highest taste scores, the cookies should be sold on day 1 or day 2 after baking. Day 3 results in a taste score that is significantly lower.

For your bakery, I recommend that you only store the cookies for 2 days in order to maintain customer satisfaction.

5.0 – RESOURCES

The annotated SAS code used to obtain the above tables and summaries is located in the appendix.

Additionally, further explanation about ANOVA, the hypotheses being tested, and pseudo-replication can be found in the appendix.

6.0 - CONSIDERATIONS

Concerns

One of the concerns that arises with this study is how the customers were chosen and how the cookies were assigned. Were the customers allowed to sample more than one cookie per day or come back the next day to sample another cookie? The model was run assuming that the customers only randomly sampled one cookie.

Another concern of the study is how the cookies were stored from day to day. Were all of the cookies stored in the same way to ensure that no other storage factor is introduced into the model. The model is run assuming that all of the batches were stored in the same manner.

Limitations

These results are based on the assumptions that a customer was only allowed to sample one cookie and all of the cookies were stored in the same manner. If these assumptions are not true than the results of this analysis may not hold true.

It was also assumed that the conditions for baking the cookies were all the same. You stated that the recipe was followed exactly other than the sugar, but other factors such as oven, baking sheets, and experience of baker were all assumed to be the same for this analysis. If there were certain variations in the baking process adding those to the analysis may be important.

Thank you for the opportunity to work with your bakery on choosing the best chocolate chip cookies for your customers. If you have any additional questions, please do not hesitate to reach out.

*Erica Junqueira
Graduate Student Consultant*

Appendix

Averaged Summarized Data: Bakery_Summary

1	1	double	1	6.8
2	1	double	4	7.4
3	1	double	7	6.4
4	1	half	2	5.2
5	1	half	5	7.0
6	1	half	8	6.2
7	1	regular	3	6.2
8	1	regular	6	6.4
9	1	regular	9	6.6
10	2	double	1	6.2
11	2	double	4	6.4
12	2	double	7	6.2
13	2	half	2	4.4
14	2	half	5	7.0
15	2	half	8	6.8
16	2	regular	3	6.8
17	2	regular	6	7.8
18	2	regular	9	6.0
19	3	double	1	5.8
20	3	double	4	4.0
21	3	double	7	4.8
22	3	half	2	5.0
23	3	half	5	4.6
24	3	half	8	5.6
25	3	regular	3	6.2
26	3	regular	6	5.8
27	3	regular	9	5.8

Numeric Summary of by sugar, day, and batch

Analysis Variable : taste											
sugar	day	batch	N Obs	N	Mean	Std Dev	Median	Lower Quartile	Upper Quartile	Minimum	Maximum
double	1	1	5	5	6.8000000	0.8366600	7.0000000	6.0000000	7.0000000	6.0000000	8.0000000
		4	5	5	7.4000000	0.8944272	8.0000000	7.0000000	8.0000000	6.0000000	8.0000000
		7	5	5	6.4000000	1.1401754	6.0000000	6.0000000	7.0000000	5.0000000	8.0000000
	2	1	5	5	6.2000000	1.3038405	6.0000000	5.0000000	7.0000000	5.0000000	8.0000000
		4	5	5	6.4000000	1.6733201	6.0000000	5.0000000	7.0000000	5.0000000	9.0000000
		7	5	5	6.2000000	0.8366600	6.0000000	6.0000000	7.0000000	5.0000000	7.0000000
	3	1	5	5	5.8000000	3.5637059	7.0000000	2.0000000	9.0000000	2.0000000	9.0000000
		4	5	5	4.0000000	1.5811388	4.0000000	3.0000000	5.0000000	2.0000000	6.0000000
		7	5	5	4.8000000	1.7888544	5.0000000	3.0000000	6.0000000	3.0000000	7.0000000
half	1	2	5	5	5.2000000	1.3038405	5.0000000	4.0000000	6.0000000	4.0000000	7.0000000
		5	5	5	7.0000000	2.4494897	6.0000000	6.0000000	9.0000000	4.0000000	10.0000000
		8	5	5	6.2000000	1.3038405	6.0000000	5.0000000	7.0000000	5.0000000	8.0000000
	2	2	5	5	4.4000000	1.9493589	5.0000000	3.0000000	5.0000000	2.0000000	7.0000000
		5	5	5	7.0000000	1.5811388	7.0000000	6.0000000	8.0000000	5.0000000	9.0000000
		8	5	5	6.8000000	1.6431677	6.0000000	6.0000000	8.0000000	5.0000000	9.0000000
	3	2	5	5	5.0000000	2.4494897	4.0000000	4.0000000	7.0000000	2.0000000	8.0000000
		5	5	5	4.6000000	1.6733201	5.0000000	4.0000000	6.0000000	2.0000000	6.0000000
		8	5	5	5.6000000	1.8165902	5.0000000	4.0000000	7.0000000	4.0000000	8.0000000
regular	1	3	5	5	6.2000000	2.2803509	6.0000000	5.0000000	6.0000000	4.0000000	10.0000000
		6	5	5	6.4000000	2.3021729	7.0000000	4.0000000	8.0000000	4.0000000	9.0000000
		9	5	5	6.6000000	1.1401754	7.0000000	6.0000000	7.0000000	5.0000000	8.0000000
	2	3	5	5	6.8000000	0.8366600	7.0000000	6.0000000	7.0000000	6.0000000	8.0000000
		6	5	5	7.8000000	1.9235384	8.0000000	7.0000000	9.0000000	5.0000000	10.0000000
		9	5	5	6.0000000	0.7071068	6.0000000	6.0000000	6.0000000	5.0000000	7.0000000
	3	3	5	5	6.2000000	0.8366600	6.0000000	6.0000000	7.0000000	5.0000000	7.0000000
		6	5	5	5.8000000	1.6431677	7.0000000	4.0000000	7.0000000	4.0000000	7.0000000
		9	5	5	5.8000000	1.6431677	6.0000000	6.0000000	7.0000000	3.0000000	7.0000000

Full Output of ANOVA Results

Model Information	
Data Set	WORK.BAKERY_SUMMARYSQ
Dependent Variable	avg_taste
Covariance Structure	Variance Components
Estimation Method	Type 3
Residual Variance Method	Factor
Fixed Effects SE Method	Model-Based
Degrees of Freedom Method	Containment

Class Level Information		
Class	Levels	Values
day	3	1 2 3
batch	9	1 2 3 4 5 6 7 8 9
sugar	3	double half regular

Dimensions	
Covariance Parameters	2
Columns in X	16
Columns in Z	9
Subjects	1
Max Obs per Subject	27

Number of Observations	
Number of Observations Read	27
Number of Observations Used	27
Number of Observations Not Used	0

Type 3 Analysis of Variance								
Source	DF	Sum of Squares	Mean Square	Expected Mean Square	Error Term	Error DF	F Value	Pr > F
sugar	2	1.905185	0.952593	Var(Residual) + 3 Var(batch(sugar)) + Q(sugar,day*sugar)	MS(batch(sugar))	6	1.32	0.3348
day	2	7.878519	3.939259	Var(Residual) + Q(day,day*sugar)	MS(Residual)	12	7.93	0.0064
day*sugar	4	1.890370	0.472593	Var(Residual) + Q(day*sugar)	MS(Residual)	12	0.95	0.4685
batch(sugar)	6	4.328889	0.721481	Var(Residual) + 3 Var(batch(sugar))	MS(Residual)	12	1.45	0.2739
Residual	12	5.964444	0.497037	Var(Residual)	-	-	-	-

Covariance Parameter Estimates	
Cov Parm	Estimate
batch(sugar)	0.07481
Residual	0.4970

Fit Statistics	
-2 Res Log Likelihood	50.6
AIC (Smaller is Better)	54.6
AICC (Smaller is Better)	55.4
BIC (Smaller is Better)	55.0

Type 3 Tests of Fixed Effects				
Effect	Num DF	Den DF	F Value	Pr > F
sugar	2	6	1.32	0.3348
day	2	12	7.93	0.0064
day*sugar	4	12	0.95	0.4685

Additional Explanation

ANOVA

The ANOVA examines the explanatory variables to compare the mean scores for the different grouping of cookies. The null hypothesis states that the means for each of the categories within a variable are equal. For example, the mean score for sugar(double) is **equal** to mean score for sugar(half) **and** sugar(regular). The alternative hypothesis states that the means for the categories are not equal. For example, the mean score for sugar(double) is **not equal** to mean score for sugar(regular) **or** sugar(half). Only one mean has to differ from the others in order to reject the null hypothesis. The null hypothesis is then evaluated at a 0.05 significance level which creates a 95% confidence level for choosing to accept or reject the null hypothesis. If the p-value for an explanatory variable is less than 0.05, we can confidently reject the null hypothesis and conclude the alternative hypothesis.

Pseudo-Replication

In this study, each batch of cookies needs to be treated as the experimental unit. Each individual cookie is the sample unit. The cookies in each batch are not independent of each other since they have many similarities. If the dependence of the cookies is not taken into consideration, it can lead to an over-inflation of the degrees of freedom. Averaging the five sampling units allows for a more precise taste score and accounts for the dependence of the taste scores.

SAS Code

```

/*****
Awesome Bakery Study
The following code analyzes the results of a study on customer taste satisfaction
for chocolate chip cookies with varying levels of sugar and freshness. The response
variable is taste and the explanatory variables are sugar and day.
*****/

```

```
Data Bakery;
```

```
input taste day sugar $ batch ;
```

```
datalines;
```

```

7      1      double 1
8      1      double 1
6      1      double 1
6      1      double 1
7      1      double 1
5      2      double 1
8      2      double 1
5      2      double 1
7      2      double 1
6      2      double 1
2      3      double 1
9      3      double 1
2      3      double 1
7      3      double 1
9      3      double 1
4      1      half    2
4      1      half    2
7      1      half    2
5      1      half    2
6      1      half    2
3      2      half    2
2      2      half    2
5      2      half    2
7      2      half    2
5      2      half    2
8      3      half    2
4      3      half    2
7      3      half    2
4      3      half    2
2      3      half    2
5      1      regular 3
6      1      regular 3
4      1      regular 3
6      1      regular 3
10     1      regular 3
7      2      regular 3
8      2      regular 3
6      2      regular 3
7      2      regular 3
6      2      regular 3
6      3      regular 3
6      3      regular 3
7      3      regular 3
5      3      regular 3
7      3      regular 3
8      1      double 4
8      1      double 4
7      1      double 4
6      1      double 4
8      1      double 4
7      2      double 4
6      2      double 4
9      2      double 4
5      2      double 4
5      2      double 4

```

6	3	double	4
5	3	double	4
2	3	double	4
4	3	double	4
3	3	double	4
6	1	half	5
10	1	half	5
4	1	half	5
9	1	half	5
6	1	half	5
9	2	half	5
5	2	half	5
8	2	half	5
6	2	half	5
7	2	half	5
5	3	half	5
6	3	half	5
6	3	half	5
4	3	half	5
2	3	half	5
7	1	regular	6
8	1	regular	6
4	1	regular	6
4	1	regular	6
9	1	regular	6
9	2	regular	6
8	2	regular	6
5	2	regular	6
10	2	regular	6
7	2	regular	6
4	3	regular	6
7	3	regular	6
7	3	regular	6
4	3	regular	6
7	3	regular	6
6	1	double	7
8	1	double	7
5	1	double	7
6	1	double	7
7	1	double	7
6	2	double	7
7	2	double	7
5	2	double	7
7	2	double	7
6	2	double	7
3	3	double	7
7	3	double	7
6	3	double	7
5	3	double	7
3	3	double	7
5	1	half	8
6	1	half	8
7	1	half	8
8	1	half	8
5	1	half	8
5	2	half	8
6	2	half	8
6	2	half	8
9	2	half	8
8	2	half	8
4	3	half	8
5	3	half	8
4	3	half	8
7	3	half	8
8	3	half	8
7	1	regular	9
6	1	regular	9
7	1	regular	9
5	1	regular	9

```

8      1      regular 9
6      2      regular 9
7      2      regular 9
6      2      regular 9
6      2      regular 9
5      2      regular 9
6      3      regular 9
7      3      regular 9
6      3      regular 9
7      3      regular 9
3      3      regular 9
;
Run;

Proc Print data=Bakery;
Run;

/*****
                        EDA
*****/

/*****DATA CLEANUP*****/
/* The following procedure steps finds the average taste score for each group of day,sugar and batch.*/

Proc SQL;
create table Bakery_summary as
select day, sugar, batch, avg(taste) as avg_taste
from Bakery
group by day, sugar , batch;
quit;

/*The following procedure step prints the summarized data containing 27 observations.*/

Proc Print data= Bakery_Summary;
Run;

/*****NUMERIC SUMMARIES *****/
/* Produces numeric summaries of the dataset Bakery_Summary.*/

Proc Means Data=Bakery_Summary n mean std median q1 q3 min max;
class sugar day ;
var avg_taste;
Run;

Proc Means Data=Bakery_Summary n mean std median q1 q3 min max; /*Displayed in appendix*/
class sugar day batch;
var avg_taste;
Run;

ods graphics on;

/*****BOXPLOTS*****/
/* The following four procedure steps create boxplots of the Bakery_Summary data.*/

Proc SGpanel data=Bakery_Summary; /*Boxplots of average taste by sugar level*/
panelby sugar /rows=1 columns=3;
vbox avg_taste;
Run;
Proc SGpanel data=Bakery_Summary; /*Boxplots of average taste by day*/
panelby day/ rows=1 columns=3;
vbox avg_taste;
Run;
Proc SGpanel data=Bakery_Summary; /*Boxplots of average taste by sugar level and batch*/
panelby sugar / rows=1 columns=3;

```



```
        vbox avg_taste / category= batch;
Run;

Proc SGpanel data=Bakery_Summary; /*Boxplots of average taste by sugar level and day*/
    panelby sugar/ rows=1 columns=3;
    vbox avg_taste/ category= day;
Run;

/*****INTERACTION PLOTS*****/
/*The following three procedure steps are used to create an interaction plot of sugar*day.*/

Proc Sort data=Bakery_Summary out=outday;
    by sugar day;
Run;

Proc Means data=outday;
    by sugar day;
    var avg_taste;
    output out=tasteday mean=mn_taste;
Run;

symbol1 value=circle color=black interpol= join;
symbol2 value=square color=black interpol=join;
symbol3 value= plus color=black interpol=join;
Proc Gplot data=tasteday; /*Interaction plot of average taste by sugar level and day*/
plot mn_taste*day=sugar;
Run;

/*****
                        STATISTICAL ANALYSIS
*****/

Proc Mixed data= Bakery_Summarysq method=type3;
    class day batch sugar;
    model avg_taste= sugar day sugar*day /residual;
    random batch(sugar);
    store outBakery;
Run;

/* Mean Comparisons for Day*/

Proc Plm restore= outBakery;
    lsmeans day / adjust=tukey plot=meanplot cl lines;
Run;
```